

Programme basic du calcul du jour julien et du temps sidéral
Tous les calculs sont faits en double précision (# accolé aux noms des variables)

```

10 '
20 ' prg TPSIDER1.BAS calcul temps sideral
30 ' a partir de la date et temps universel
40 ' Ph Merlin 07/08/1989
50 '
60 ' variables et constantes sont en doubles precision
70 DEFDBL A-H,O-Z
80 C0 = 24110.54841#
90 C1 = 8640184.812866#
100 C2 = .093104#
110 C3 = -.0000062#
120 '
130 ' mettre ici la longitude du lieu en heures décimales
140 ' longitude 0 = temps sideral a Greenwich
150 '
160 LONG = 0.319033# : ' Observ = .319033#
170 CLS
180 LOCATE 13,27,0:PRINT" T E M P S   S I D E R A L"
190 LOCATE 22,20,0:PRINT"appuyer sur une touche (f ou F pour finir)"
200 L$=INKEY$: IF L$="" GOTO 200 : ' attente d'un caractere pour continuer
210 CLS
220 IF L$="f" OR L$="F" THEN END : ' test d'arret si f ou F appuyes
230 '
240 ' entree de la date
250 '
260 LOCATE 10,1
270 PRINT " Date (AAAA.MMJJHHMMSS) "
280 PRINT " retour = date et heure de l'ordinateur : "
290 '
300 LINE INPUT L$
310 IF LEN(L$) = 0 THEN GOSUB 840 : ' recuperation date et heure de l'ordinateur
320 ' extraction des donnees
330 AA$ = MID$(L$,1,4) : AA = VAL(AA$) : ' annee
340 MM$ = MID$(L$,6,2) : MM = VAL(MM$) : ' mois
350 JO$ = MID$(L$,8,2) : JO = VAL(JO$) : ' jour
360 HH$ = MID$(L$,10,2) : HH = VAL(HH$) : ' heure
370 MI$ = MID$(L$,12,2) : MI = VAL(MI$) : ' minute
380 SS$ = MID$(L$,14,2) : SS = VAL(SS$) : ' seconde
390 '
400 ' ----- calcul Jour Julien -----
410 '
420 PRINT AA MM JO HH MI SS
430 DT = AA+MM/100+JO/10000 : ' annee decimale pour test periode gregorienne
440 HD = HH+MI/60+SS/3600 : ' heure decimale
450 Y = AA
460 M=MM
470 IF M <= 2 THEN Y = Y-1 : M = M+12
480 IF DT <= 1582.1015# THEN B = 0 : GOTO 510
490 A = FIX(Y/100#)
500 B = 2 - FIX(Y/100#) + FIX(Y/400#)
510 AJJ = FIX(365.25#* Y) + FIX(30.6# * (M+1)) + JO + 1720994.5# + B
520 PRINT "      JJ (à 0h TU) = ";AJJ
530 TUD = (AJJ - 2451545#)/36525# : ' temps en siecles julien
540 '
550 ' calcul temps sideral
560 '
570 TSMG = (((C3 * TUD) + C2) * TUD) + C1) * TUD + C0
580 TSMG = TSMG /3600#
590 TSMG = TSMG /24#
600 '
610 ' reduction a un jour
620 '
630 IF TSMG > 1# THEN TSMG = TSMG - 1# : GOTO 630
640 IF TSMG < 0# THEN TSMG = TSMG + 1# : GOTO 640
650 PRINT "      TS GM 0h ";TSMG * 24 : ' en heures decimales
660 IF INKEY$ = "" GOTO 660
670 '
680 ' temps sideral local
690 '
700 TSML = TSMG * 24# + LONG + 1.00273790934# * HD
710 IF TSML > 24# THEN TSML = TSML - 24#
720 '
730 ' mise en forme resultat
740 '
750 H$ = STR$(INT(TSML))+ "h"
760 T = (TSML - INT(TSML)) * 60#
770 H$ = H$ + STR$(INT(T))+ "mn"
780 T = (INT(((T-INT(T)) * 60#)*1000#))/1000#
790 H$ = H$ + STR$(T)+ "s"

```

```
800 LOCATE 18,15,0:PRINT "Temps sideral (";L$;" ) : ";H$
810 IF INKEY$ = "" GOTO 810
820 CLS
830 END
840 '
850 ' date et temps de l'ordinateur
860 ' attention aux systemes US qui donnent le mois avant le jour
870 '
880 T$ = DATE$ : TT$ = TIME$
890 L$ = MID$(T$,7,4)+". "+MID$(T$,1,2)+MID$(T$,4,2)
900 L$ = L$ + MID$(TT$,1,2)+MID$(TT$,4,2)+MID$(TT$,7,2)
910 RETURN
```

Philippe Merlin Observatoire de Lyon

http://www-obs.univ-lyon1.fr/labo/fc/cdroms/docu_astro/jour_julien/cal_tuts.pdf